Pensieve header: The double and meta-double of the 2D pencil; continues pensieve://2017-05/.

Issues:

1. S does not invert R. (Perhaps because H must be interpreted as $e^{\hbar h}$).
2. dm is not meta-associative.
3. R doesn't satisfy YB.

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\2017-06"];
```

The "degree carrier" is $\hbar$, and all "coupling constants" are proportional to it.

```
$TD = 3; ℏ /: ℏ^d_. /; d > $TD := 0;
```

The 2D Lie BiAlgebra Pencil

We hope to stick to $A = e^{\hbar \beta a}$ and to $B = e^{\hbar \alpha b}$, where $[a, x] = \alpha x$ and $[b, y] = -\beta y$.
Also, $\Delta_{12}(a, A, x, b, B, y) = (a_1 + a_2,\ A_1 A_2,\ x_1 + A_1 x_2,\ b_1 + b_2,\ B_1 B_2,\ y_1 B_2 + y_2)$.
Also, (a, x) and $\hbar(b, y)$ are dual bases.

```
AlgebraAtom = a | A[_] | x | b | B[_] | y;
$PBWRule = {A[_] → 1, a → 2, x → 3, B[_] → 4, b → 5, y → 6};
```

```
B[a, x] = α x; B[x, A[n_]] = (e^(-n ℏ α β) - 1) U[A[n], x]; B[a, A[_]] = 0;
B[b, y] = -β y; B[y, B[n_]] = (e^(n ℏ α β) - 1) U[B[n], y]; B[b, B[_]] = 0;
```

```
ExpandAB[ℰ_] := Expand@Normal@Series[ℰ //. {
    c_. U_i_[λ___, A[n_], ρ___] ⧴
     Expand[c Sum[(ℏ^d β^d n^d)/d! U_i[λ, Sequence @@ Table[a, {d}], ρ], {d, 0, $TD}]],
    c_. U_i_[λ___, B[n_], ρ___] ⧴ Expand[
      c Sum[(ℏ^d α^d n^d)/d! U_i[λ, Sequence @@ Table[b, {d}], ρ], {d, 0, $TD}]]
   },
   {ℏ,
    0,
    $TD}]
```

UEA with provisional modification

This section is based on penseive://Projects/UEA/.

```
B[0, _] = 0; B[_, 0] = 0;
B[c_ * x : AlgebraAtom, y_] := Expand[c B[x, y]];
B[y_, c_ * x : AlgebraAtom] := Expand[c B[y, x]];
B[x_Plus, y_] := B[#, y] & /@ x;
B[x_, y_Plus] := B[x, #] & /@ y;
B[x_, x_] = 0;
B[y_, x_] := Expand[-B[x, y]];
```

```
x_ ≤ y_ := OrderedQ[{x, y} /. $PBWRule]; x_ < y_ := ! OrderedQ[{y, x} /. $PBWRule];
UU_i_[] := U_i[]; UU_i_[1] := U_i[];
UU_i_[x_[n_]^p_] := U_i[x[n p]];
UU_i_[x_^p_] := UU_i @@ Table[x, {p}];
UU_i_[ℰ_] := ℰ /. {
    U[xs__] ⧴ U_i[xs],
    x : AlgebraAtom ⧴ U_i[x]
   };
UU_i_[x_, xs__] := UU_t1[x] UU_t2[xs] // Expand // m_{t1,t2→i};
USimp[ℰ_] := Collect[ℰ, Times[U_[__] ..], Expand];
USimp[ℰ_] := Expand[ℰ];
```

```
m_s__[0] = 0;
m_s__[x_Plus] := m_s /@ x;
m_{i_→j_}[ℰ_] := ℰ /. U_i → U_j;
```

```
m_{i_,j_→k_}[c_. U_i_[x___] U_j_[]] := c U_k[x];
m_{i_,j_→k_}[c_. U_i_[] U_j_[y___]] := c U_k[y];
m_{i_,j_→k_}[c_. U_i_[xx___, x_[n1_]] U_j_[x_[n2_], yy___]] :=
  USimp[c If[n1 + n2 == 0, U_i[xx] U_j[yy], U_i[xx, x[n1 + n2]] U_j[yy]] // m_{i,j→k}];
m_{i_,j_→k_}[c_. U_i_[xx___, x_] U_j_[y_, yy___]] := If[x ≤ y,
   c U_k[xx, x, y, yy],
   ((U_i[xx] (U_j[y, x] + UU_j[B[x, y]]) // Expand // m_{i,j→i}) U_j[yy] // Expand // m_{i,j→k})
     c // USimp
  ];
```

```
Supp[ℰ_] := Union@Cases[{ℰ}, U_i_[___] ⧴ i, ∞];
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
x_ ** y_ := Module[
   {Sx = Supp[x], Sy = Supp[y], is, σ, z},
   If[MatchQ[Sx ⋃ Sy, {_Integer ...}] && Min[Sx ⋃ Sy] < 0,
    is = Abs[Sx] ⋂ Abs[Sy];
    z = x; Do[z = m_{-i→-σ@i}[m_{i→σ@i}[z]], {i, is}];
    z = USimp[y z]; Do[z = dm_{σ@i,i→i}[z], {i, is}];
    z,
    (* else *) is = Sx ⋂ Sy;
    z = x; Do[z = m_{i→σ@i}[z], {i, is}];
    z = USimp[y z]; Do[z = m_{σ@i,i→i}[z], {i, is}];
    z
   ]
  ];
UB[x_, y_] := USimp[x ** y - y ** x];
```

```
𝕆[func_] := Normal@Series[func, {ℏ, 0, $TD}];
𝕆[specs_, func_] := Module[{rules, vars, elems},
  rules = Union @@ Cases[{specs}, U_[u___] :> Cases[{u}, r_Rule], ∞];
  vars = First /@ rules; elems = Last /@ rules;
  USimp@Total[CoefficientRules[𝕆[func], vars] /. (ps_ → c_) :> c (
        specs /. MapThread[((#1 → _) :> #3^#2) &, {vars, ps, elems}] /. U_i_ :> UU_i
       )]
 ]
```

The 2D Lie BiAlgebra Pencil, Testing

```
𝕆[U₁[a → a], e^(ℏ β a)]
```

$U_1[\,] + \beta\hbar\, U_1[a] + \frac{1}{2}\beta^2\hbar^2 U_1[a, a] + \frac{1}{6}\beta^3\hbar^3 U_1[a, a, a]$

```
USimp@With[{A = 𝕆[U₁[a → a], e^(n ℏ β a)]}, UB[U₁[x], A] - 𝕆[e^(-n ℏ α β) - 1] A ** U₁[x]]
```

0

```
B[x, A[3]]
```

$\left(-1 + e^{-3\alpha\beta\hbar}\right) U[A[3], x]$

```
$TD = 6;
USimp@With[{Bn = 𝕆[U₁[b → b], e^(n ℏ α b)]}, UB[U₁[y], Bn] - 𝕆[e^(n ℏ α β) - 1] Bn ** U₁[y]]
```

0

```
z = U₁[a, A[2], x, x, x] U₂[a, a, x] U₃[a, a, A[-3], x];
(z // m_{1,2→1} // m_{1,3→1}) - (z // m_{2,3→2} // m_{1,2→1})
```

0

```
z = U1[b, B[2], y, y, y] U2[b, b, y] U3[b, b, B[-3], y];
(z // m1,2→1 // m1,3→1) - (z // m2,3→2 // m1,2→1)
```

0

The Co-Product and Co-Associativity

```
Δi_→j_,k_[ℰ_] := USimp@Module[{tj, tk}, ℰ /. {
    Ui[] → Uj[] Uk[],
    Ui[a, xs___] :→
     (USimp[(Uj[a] Uk[] + Uj[] Uk[a]) Δi→tj,tk[Ui[xs]]] // mj,tj→j // mk,tk→k),
    Ui[A[n_], xs___] :→ (USimp[Uj[A[n]] Uk[A[n]] Δi→tj,tk[Ui[xs]]] //
       mj,tj→j // mk,tk→k),
    Ui[x, xs___] :→ (USimp[(Uj[x] Uk[A[1]] + Uj[] Uk[x]) Δi→tj,tk[Ui[xs]]] //
       mj,tj→j // mk,tk→k),
    Ui[b, xs___] :→ (USimp[(Uj[b] Uk[] + Uj[] Uk[b]) Δi→tj,tk[Ui[xs]]] //
       mj,tj→j // mk,tk→k),
    Ui[B[n_], xs___] :→ (USimp[Uj[B[n]] Uk[B[n]] Δi→tj,tk[Ui[xs]]] //
       mj,tj→j // mk,tk→k),
    Ui[y, xs___] :→ (USimp[(Uj[y] Uk[] + Uj[B[1]] Uk[y]) Δi→tj,tk[Ui[xs]]] //
       mj,tj→j // mk,tk→k)
   }];
```

```
Δi_→j_,k_,l__[ℰ_] := ℰ // Δi→j,k // Δk→k,l
```

```
U1[x] // Δ1→1,2
```

$U_1[]\,U_2[x] + U_1[x]\,U_2[A[1]]$

```
{lhs = U1[x] // Δ1→1,2 // Δ2→2,3, rhs = U1[x] // Δ1→1,3 // Δ1→1,2, lhs == rhs}
```

$\{U_1[]\,U_2[]\,U_3[x] + U_1[]\,U_2[x]\,U_3[A[1]] + U_1[x]\,U_2[A[1]]\,U_3[A[1]],$
$U_1[]\,U_2[]\,U_3[x] + U_1[]\,U_2[x]\,U_3[A[1]] + U_1[x]\,U_2[A[1]]\,U_3[A[1]],\ \text{True}\}$

```
U1[y] // Δ1→1,2
```

$U_1[y]\,U_2[] + U_1[B[1]]\,U_2[y]$

```
{lhs = U1[y] // Δ1→1,2 // Δ2→2,3, rhs = U1[y] // Δ1→1,3 // Δ1→1,2, lhs == rhs}
```

$\{U_1[y]\,U_2[]\,U_3[] + U_1[B[1]]\,U_2[y]\,U_3[] + U_1[B[1]]\,U_2[B[1]]\,U_3[y],$
$U_1[y]\,U_2[]\,U_3[] + U_1[B[1]]\,U_2[y]\,U_3[] + U_1[B[1]]\,U_2[B[1]]\,U_3[y],\ \text{True}\}$

```
z = U1[a, A[2], x, x, x] U2[a, a, A[-3], x];
(z // m1,2→1 // Δ1→1,2) - (z // Δ2→3,4 // Δ1→1,2 // m1,3→1 // m2,4→2)
```

0

```
z = U1[b, B[2], y, y, y] U2[b, b, B[-3], y];
(z // m1,2→1 // Δ1→1,2) - (z // Δ2→3,4 // Δ1→1,2 // m1,3→1 // m2,4→2)
```

0

The Antipode

```
S[a] = -a; S[A[n_]] := A[-n]; S[x] = -e^(ħ α β) U[A[-1], x];
S[b] = -b; S[B[n_]] := B[-n]; S[y] = -U[B[-1], y];
```

```
S_i_[ℰ_] := Module[{ti}, USimp[
    ℰ /. U_i[x_, xs___] :> m_{ti,i→i}[Expand[UU_i[S[x]] S_ti[U_ti[xs]]]]
  ]];
```

```
{lhs = S_1[U_1[x]], rhs = -U_1[x] ** U_1[A[-1]], lhs == rhs}
```

$\{-e^{\alpha\beta\hbar} U_1[A[-1], x], -e^{\alpha\beta\hbar} U_1[A[-1], x], \text{True}\}$

```
{U_1[x] // S_1 // S_1, U_1[y] // S_1 // S_1}
```

$\{e^{\alpha\beta\hbar} U_1[x], e^{\alpha\beta\hbar} U_1[y]\}$

```
{z = U_1[]; (z // Δ_{1→1,2} // S_1 // m_{1,2→1}), z = U_1[]; (z // Δ_{1→1,2} // S_2 // m_{1,2→1})}
```

$\{U_1[], U_1[]\}$

```
z = U_1[a, A[3], x, x];
{z // Δ_{1→1,2} // S_2 // m_{1,2→1}, z // Δ_{1→1,2} // S_2 // m_{2,1→1},
 z // Δ_{1→1,2} // S_1 // m_{1,2→1}, z // Δ_{1→1,2} // S_1 // m_{2,1→1}}
```

$\{0, 0, 0, 2 e^{6\alpha\beta\hbar} \alpha U_1[x, x] - 2 e^{7\alpha\beta\hbar} \alpha U_1[x, x] - 2 e^{8\alpha\beta\hbar} \alpha U_1[x, x] + 2 e^{9\alpha\beta\hbar} \alpha U_1[x, x]\}$

```
z = U_1[b, B[-3], y, y, y];
{z // Δ_{1→1,2} // S_2 // m_{1,2→1}, z // Δ_{1→1,2} // S_2 // m_{2,1→1},
 z // Δ_{1→1,2} // S_1 // m_{1,2→1}, z // Δ_{1→1,2} // S_1 // m_{2,1→1}}
```

$\{0, 0, 0, -3 \beta U_1[B[-3], y, y, y] + 3 e^{\alpha\beta\hbar} \beta U_1[B[-3], y, y, y] + 3 e^{2\alpha\beta\hbar} \beta U_1[B[-3], y, y, y] - 3 e^{4\alpha\beta\hbar} \beta U_1[B[-3], y, y, y] - 3 e^{5\alpha\beta\hbar} \beta U_1[B[-3], y, y, y] + 3 e^{6\alpha\beta\hbar} \beta U_1[B[-3], y, y, y]\}$

```
{z = U_1[a, A[2], x, x, x] U_2[a, a, A[-3], x]; (z // m_{1,2→1} // S_1) - (z // S_1 // S_2 // m_{2,1→1}),
 z = U_1[b, B[2], y, y, y] U_2[b, b, b, B[6], y]; (z // m_{1,2→1} // S_1) - (z // S_1 // S_2 // m_{2,1→1})}
```

$\{0, 0\}$

The Pairing

```
P[U[], U[B[_]]] = P[U[A[_]], U[]] = 1;
P[U[], U[__]] = P[U[__], U[]] = 0;
( P[U[a], U[b]] = ħ^-1          P[U[a], U[B[n_]]] = n α              P[U[a], U[y]] = 0
  P[U[A[n_]], U[b]] = n β      P[U[A[n_]], U[B[m_]]] = e^(n m ħ α β)  P[U[A[_]], U[y]] = 0
  P[U[x], U[b]] = 0             P[U[x], U[B[_]]] = 0                  P[U[x], U[y]] = ħ^-1 );
```

```
P[U[], U[]] = 1;
P_i_,j_[ℰ_] := USimp[ℰ /. U_i[xs___] U_j[ys___] → P[U[xs], U[ys]]];
```

The pairing sequence: ⟨one,one⟩ (above), ⟨many,one⟩, ⟨many,many⟩.

```
P[U[x_, xs__], U[y_]] := P[U[x, xs], U[y]] =
   Module[{i, j, k, l}, USimp[U_i[x] U_j[xs] Δ_{k→k,l}[U_k[y]]] // P_{i,k} // P_{j,l}];
P[U[xs__], U[y_, ys__]] := P[U[xs], U[y, ys]] =
   Module[{i, j, k, l}, USimp[Δ_{i→i,j}[U_i[xs]] U_k[y] U_l[ys]] // P_{i,k} // P_{j,l}];
```

```
z = U_i[a] U_j[x] U_k[y];
{m_{i,j→i}[z] - m_{j,i→i}[z], Δ_{k→k,l}[z] - Δ_{k→l,k}[z]}
```

{α U_i[x] U_k[y], U_i[a] U_j[x] U_k[y] U_l[] - U_i[a] U_j[x] U_k[] U_l[y] +
  U_i[a] U_j[x] U_k[B[1]] U_l[y] - U_i[a] U_j[x] U_k[y] U_l[B[1]]}

```
Table[z = U_i[xi] U_j[xj] U_k[yk];
 {(m_{i,j→i}[z] - m_{j,i→i}[z]) // P_{i,k}, (Δ_{k→k,l}[z] - Δ_{k→l,k}[z]) // P_{i,k} // P_{j,l}},
 {xi, {a, x}}, {xj, {a, x}}, {yk, {b, y}}]
```

$$\left\{\left\{\{\{0,0\},\{0,0\}\},\left\{\{0,0\},\left\{\frac{\alpha}{\hbar},\frac{\alpha}{\hbar}\right\}\right\}\right\},\left\{\left\{\{0,0\},\left\{-\frac{\alpha}{\hbar},-\frac{\alpha}{\hbar}\right\}\right\},\{\{0,0\},\{0,0\}\}\right\}\right\}$$

```
Table[z = U_i[xi] U_k[yk] U_l[yl];
 {(Δ_{i→i,j}[z] - Δ_{i→j,i}[z]) // P_{i,k} // P_{j,l}, (m_{k,l→k}[z] - m_{l,k→k}[z]) // P_{i,k}},
 {xi, {a, x}}, {yk, {b, y}}, {yl, {b, y}}]
```

$$\left\{\left\{\{\{0,0\},\{0,0\}\},\{\{0,0\},\{0,0\}\}\right\},\left\{\left\{\{0,0\},\left\{-\frac{\beta}{\hbar},-\frac{\beta}{\hbar}\right\}\right\},\left\{\left\{\frac{\beta}{\hbar},\frac{\beta}{\hbar}\right\},\{0,0\}\right\}\right\}\right\}$$

```
lhs = Factor@Table[ℏ^n P[U @@ Table[x, {n}], U @@ Table[y, {n}]], {n, $TD = 7}]
```

$$\{1,\ 1+e^{\alpha\beta\hbar},\ (1+e^{\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}),\ (1+e^{\alpha\beta\hbar})^2(1+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}),$$
$$(1+e^{\alpha\beta\hbar})^2(1+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}),$$
$$(1+e^{\alpha\beta\hbar})^3(1+e^{2\alpha\beta\hbar})(1-e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})^2(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}),$$
$$(1+e^{\alpha\beta\hbar})^3(1+e^{2\alpha\beta\hbar})(1-e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})^2$$
$$(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}+e^{5\alpha\beta\hbar}+e^{6\alpha\beta\hbar})\}$$

```
rhs = Simplify@FunctionExpand@Table[QFactorial[n, e^(ℏ α β)], {n, $TD = 7}]
```

$$\{1,\ 1+e^{\alpha\beta\hbar},\ (1+e^{\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}),\ (1+e^{\alpha\beta\hbar})^2(1+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}),$$
$$(1+e^{\alpha\beta\hbar})^2(1+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}),$$
$$(1+e^{\alpha\beta\hbar})^3(1+e^{2\alpha\beta\hbar})(1-e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})^2(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}),$$
$$(1+e^{\alpha\beta\hbar})^3(1+e^{2\alpha\beta\hbar})(1-e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})^2$$
$$(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}+e^{5\alpha\beta\hbar}+e^{6\alpha\beta\hbar})\}$$

```
MapThread[Equal, {lhs, rhs}]
```

{True, True, True, True, True, True, True}

```
P[U[a, a, a, a, a], U[b, b, b, b, b]]
```

$$\frac{120}{\hbar^5}$$

```
P[U[a, a, a, a, a, x, x, x, x], U[b, b, b, b, b, y, y, y, y]] // Factor
```

$$\frac{120 \left(1+e^{\alpha \beta \hbar}\right)^2 \left(1+e^{2 \alpha \beta \hbar}\right) \left(1+e^{\alpha \beta \hbar}+e^{2 \alpha \beta \hbar}\right)}{\hbar^9}$$

```
z = U₁[a, A[-3], x, x] U₂[a, a, A[1], x] U₃[b, B[2], y, y, y];
(z // m_{1,2→1} // P_{1,3}) - (z // Δ_{3→3,4} // P_{1,3} // P_{2,4})
```

0

```
z = U₁[b, B[-3], y, y] U₂[b, b, B[1], y] U₃[a, A[2], x, x, x];
(z // m_{1,2→1} // P_{3,1}) - (z // Δ_{3→3,4} // P_{3,1} // P_{4,2})
```

0

```
z = U₁[a, a, A[-3], x, x, x] U₂[b, b, B[2], y, y, y];
(z // S₁ // P_{1,2}) - (z // S₂ // P_{1,2})
```

0

The Double

```
dm_{i_,j_→k_}[ℰ_] := Module[{t1, t2, t3, h1, h2, h3},
  ℰ // Δ_{i→h1,h2,h3} // S_{h1} // Δ_{-j→t1,t2,t3} // P_{h1,t1} // P_{h3,t3} // m_{j,h2→k} // m_{-i,t2→-k}];
dΔ_{i_→j_,k_}[ℰ_] := ℰ // Δ_{i→j,k} // Δ_{-i→-j,-k};
```

```
U₋₁[] U₁[a] U₋₂[b] U₂[] // dm_{1,2→1}
```

$U_{-1}[b]\, U_1[a]$

```
{U₋₁[] U₁[a] U₋₂[y] U₂[] // dm_{2,1→1}, U₋₁[] U₁[a] U₋₂[y] U₂[] // dm_{1,2→1}}
```

$\{U_{-1}[y]\, U_1[a],\ -\alpha\, U_{-1}[y]\, U_1[\,] + U_{-1}[y]\, U_1[a]\}$

```
U₋₁[] U₁[A[1]] U₋₂[y] U₂[] // dm_{1,2→1}
```

$e^{-\alpha \beta \hbar}\, U_{-1}[y]\, U_1[A[1]]$

```
U₋₁[] U₁[x] U₋₂[b] U₂[] // dm_{1,2→1}
```

$\beta\, U_{-1}[\,]\, U_1[x] + U_{-1}[b]\, U_1[x]$

```
U₋₁[] U₁[x] U₋₂[B[1]] U₂[] // dm_{1,2→1}
```

$e^{\alpha \beta \hbar}\, U_{-1}[B[1]]\, U_1[x]$

```
U₋₁[] U₁[x] U₋₂[y] U₂[] // dm_{1,2→1}
```

$$\frac{U_{-1}[B[1]]\, U_1[\,]}{\hbar} + U_{-1}[y]\, U_1[x] - \frac{U_{-1}[\,]\, U_1[A[1]]}{\hbar}$$

```
z = U₋₁[] U₁[x] U₋₂[y] U₂[] U₋₃[b] U₃[];
(z // dm_{1,2→1} // dm_{1,3→1}) - (z // dm_{2,3→2} // dm_{1,2→1})
```

0

```
z = U₋₁[b, b, y] U₁[A[2], x, x] U₋₂[B[-1], y, y] U₂[a] U₋₃[b, y] U₃[a, a, x];
(z // dm_{1,2→1} // dm_{1,3→1}) - (z // dm_{2,3→2} // dm_{1,2→1})
```

0

The R-Matrix

Using Quesne's formula.

```
(*R_{i_,j_} :=𝕆[
  U_{-i}[y→y, b→b] U_i[] U_{-j}[] U_j[a→a, x→x],
  Series[Exp[ℏ b a + Sum_{k=1}^{$TD} (1-e^{ℏ α β})^k (ℏ x y)^k / (k (1-e^{k ℏ α β}))], {ℏ,0,$TD}]
 ]*)
```

```
R_{i_,j_} := 𝕆[
  U_{-i}[b → b, y → y] U_i[] U_{-j}[] U_j[a → a, x → x],
  Series[Exp[ℏ b a + Sum_{k=1}^{$TD} (1 - e^{ℏ α β})^k (ℏ x y)^k / (k (1 - e^{k ℏ α β}))], {ℏ, 0, $TD}]
 ]
```

```
$TD = 2; R_{1,2}
```

$U_{-2}[\,]\,U_{-1}[\,]\,U_1[\,]\,U_2[\,] + \hbar\,U_{-2}[\,]\,U_{-1}[b]\,U_1[\,]\,U_2[a] + \hbar\,U_{-2}[\,]\,U_{-1}[y]\,U_1[\,]\,U_2[x] + \frac{1}{2}\hbar^2\,U_{-2}[\,]\,U_{-1}[b,b]\,U_1[\,]\,U_2[a,a] + \hbar^2\,U_{-2}[\,]\,U_{-1}[b,y]\,U_1[\,]\,U_2[a,x] + \frac{1}{2}\hbar^2\,U_{-2}[\,]\,U_{-1}[y,y]\,U_1[\,]\,U_2[x,x]$

```
P[U[x, x], U[y, y]]
```

$\frac{1}{\hbar^2} + \frac{e^{\alpha\beta\hbar}}{\hbar^2}$

```
$TD = 2; {lhs = R_{1,3} // dΔ_{1→1,2} // ExpandAB,
  rhs = R_{2,3} ** R_{1,3} // ExpandAB, Coefficient[lhs - rhs, ℏ^2]} // Last
```

$\$TD = 2;\ \{lhs = R_{1,3} \mathbin{//} d\Delta_{1\to 1,2},\ rhs = R_{2,3} ** R_{1,3},\ lhs - rhs\}$

$$
\begin{aligned}
\{&U_{-3}[]\,U_{-2}[]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[] + \hbar\,U_{-3}[]\,U_{-2}[b]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[a] +\\
&\hbar\,U_{-3}[]\,U_{-2}[]\,U_{-1}[b]\,U_1[]\,U_2[]\,U_3[a] + \hbar\,U_{-3}[]\,U_{-2}[]\,U_{-1}[y]\,U_1[]\,U_2[]\,U_3[x] +\\
&\hbar\,U_{-3}[]\,U_{-2}[y]\,U_{-1}[B[1]]\,U_1[]\,U_2[]\,U_3[x] + \frac{1}{2}\hbar^2\,U_{-3}[]\,U_{-2}[b, b]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[a, a] +\\
&\hbar^2\,U_{-3}[]\,U_{-2}[b]\,U_{-1}[b]\,U_1[]\,U_2[]\,U_3[a, a] + \frac{1}{2}\hbar^2\,U_{-3}[]\,U_{-2}[]\,U_{-1}[b, b]\,U_1[]\,U_2[]\,U_3[a, a] +\\
&\hbar^2\,U_{-3}[]\,U_{-2}[b]\,U_{-1}[y]\,U_1[]\,U_2[]\,U_3[a, x] + \hbar^2\,U_{-3}[]\,U_{-2}[b, y]\,U_{-1}[B[1]]\,U_1[]\,U_2[]\,U_3[a, x] +\\
&\hbar^2\,U_{-3}[]\,U_{-2}[]\,U_{-1}[b, y]\,U_1[]\,U_2[]\,U_3[a, x] + \hbar^2\,U_{-3}[]\,U_{-2}[y]\,U_{-1}[B[1], b]\,U_1[]\,U_2[]\,U_3[a, x] +\\
&\frac{1}{2}\hbar^2\,U_{-3}[]\,U_{-2}[y, y]\,U_{-1}[B[2]]\,U_1[]\,U_2[]\,U_3[x, x] +\\
&\frac{1}{2}\hbar^2\,U_{-3}[]\,U_{-2}[]\,U_{-1}[y, y]\,U_1[]\,U_2[]\,U_3[x, x] + \frac{1}{2}\hbar^2\,U_{-3}[]\,U_{-2}[y]\,U_{-1}[B[1], y]\,U_1[]\,U_2[]\,U_3[x, x] +\\
&\frac{1}{2}e^{\alpha\beta\hbar}\hbar^2\,U_{-3}[]\,U_{-2}[y]\,U_{-1}[B[1], y]\,U_1[]\,U_2[]\,U_3[x, x],\\
&U_{-3}[]\,U_{-2}[]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[] + \hbar\,U_{-3}[]\,U_{-2}[b]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[a] +\\
&\hbar\,U_{-3}[]\,U_{-2}[]\,U_{-1}[b]\,U_1[]\,U_2[]\,U_3[a] + \hbar\,U_{-3}[]\,U_{-2}[y]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[x] +\\
&\hbar\,U_{-3}[]\,U_{-2}[]\,U_{-1}[y]\,U_1[]\,U_2[]\,U_3[x] - \alpha\hbar^2\,U_{-3}[]\,U_{-2}[b]\,U_{-1}[y]\,U_1[]\,U_2[]\,U_3[x] +\\
&\frac{1}{2}\hbar^2\,U_{-3}[]\,U_{-2}[b, b]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[a, a] + \hbar^2\,U_{-3}[]\,U_{-2}[b]\,U_{-1}[b]\,U_1[]\,U_2[]\,U_3[a, a] +\\
&\frac{1}{2}\hbar^2\,U_{-3}[]\,U_{-2}[]\,U_{-1}[b, b]\,U_1[]\,U_2[]\,U_3[a, a] + \hbar^2\,U_{-3}[]\,U_{-2}[b, y]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[a, x] +\\
&\hbar^2\,U_{-3}[]\,U_{-2}[y]\,U_{-1}[b]\,U_1[]\,U_2[]\,U_3[a, x] + \hbar^2\,U_{-3}[]\,U_{-2}[b]\,U_{-1}[y]\,U_1[]\,U_2[]\,U_3[a, x] +\\
&\hbar^2\,U_{-3}[]\,U_{-2}[]\,U_{-1}[b, y]\,U_1[]\,U_2[]\,U_3[a, x] + \frac{1}{2}\hbar^2\,U_{-3}[]\,U_{-2}[y, y]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[x, x] +\\
&\hbar^2\,U_{-3}[]\,U_{-2}[y]\,U_{-1}[y]\,U_1[]\,U_2[]\,U_3[x, x] + \frac{1}{2}\hbar^2\,U_{-3}[]\,U_{-2}[]\,U_{-1}[y, y]\,U_1[]\,U_2[]\,U_3[x, x],\\
&-\hbar\,U_{-3}[]\,U_{-2}[y]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[x] + \alpha\hbar^2\,U_{-3}[]\,U_{-2}[b]\,U_{-1}[y]\,U_1[]\,U_2[]\,U_3[x] +\\
&\hbar\,U_{-3}[]\,U_{-2}[y]\,U_{-1}[B[1]]\,U_1[]\,U_2[]\,U_3[x] - \hbar^2\,U_{-3}[]\,U_{-2}[b, y]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[a, x] -\\
&\hbar^2\,U_{-3}[]\,U_{-2}[y]\,U_{-1}[b]\,U_1[]\,U_2[]\,U_3[a, x] + \hbar^2\,U_{-3}[]\,U_{-2}[b, y]\,U_{-1}[B[1]]\,U_1[]\,U_2[]\,U_3[a, x] +\\
&\hbar^2\,U_{-3}[]\,U_{-2}[y]\,U_{-1}[B[1], b]\,U_1[]\,U_2[]\,U_3[a, x] - \frac{1}{2}\hbar^2\,U_{-3}[]\,U_{-2}[y, y]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[x, x] -\\
&\hbar^2\,U_{-3}[]\,U_{-2}[y]\,U_{-1}[y]\,U_1[]\,U_2[]\,U_3[x, x] + \frac{1}{2}\hbar^2\,U_{-3}[]\,U_{-2}[y, y]\,U_{-1}[B[2]]\,U_1[]\,U_2[]\,U_3[x, x] +\\
&\frac{1}{2}\hbar^2\,U_{-3}[]\,U_{-2}[y]\,U_{-1}[B[1], y]\,U_1[]\,U_2[]\,U_3[x, x] +\\
&\frac{1}{2}e^{\alpha\beta\hbar}\hbar^2\,U_{-3}[]\,U_{-2}[y]\,U_{-1}[B[1], y]\,U_1[]\,U_2[]\,U_3[x, x]\}
\end{aligned}
$$

$\$TD = 2;$
$\{lhs = R_{1,3} \mathbin{//} d\Delta_{1\to 1,2} \mathbin{//} ExpandAB,\ rhs = (R_{2,3}\,R_{1,4}) \mathbin{//} Expand \mathbin{//} m_{4,3\to 3} \mathbin{//} m_{-3,-4\to -3} \mathbin{//} ExpandAB,$
$\quad Coefficient[lhs - rhs, \hbar^2]\} \mathbin{//} Last$

$$\alpha\,U_{-3}[]\,U_{-2}[y]\,U_{-1}[b]\,U_1[]\,U_2[]\,U_3[x] + \alpha\,U_{-3}[]\,U_{-2}[b]\,U_{-1}[y]\,U_1[]\,U_2[]\,U_3[x]$$

```
$TD = 5;
{lhs = ExpandAB[R_{1,2} ** R_{1,3} ** R_{2,3}],
  rhs = ExpandAB[R_{2,3} ** R_{1,3} ** R_{1,2}], Coefficient[lhs - rhs, ħ^3]} // Last
```

$\alpha^2 U_{-3}[] U_{-2}[b, b] U_{-1}[y] U_1[] U_2[] U_3[x] - \beta^2 U_{-3}[] U_{-2}[] U_{-1}[y] U_1[] U_2[a, a] U_3[x]$

```
$TD = 2; (R_{1,2} R_{3,4} // S_{-1} // m_{1,3→1} // m_{2,4→2} // m_{-3,-1→-1} // m_{-4,-2→-2}) // ExpandAB
```

$U_{-2}[] U_{-1}[] U_1[] U_2[] + 2 \alpha \hbar^2 U_{-2}[] U_{-1}[b, y] U_1[] U_2[x] + \frac{1}{2} \hbar^2 U_{-2}[] U_{-1}[y, y] U_1[] U_2[x, x] - \frac{1}{2} e^{-\alpha \beta \hbar} \hbar^2 U_{-2}[] U_{-1}[y, y] U_1[] U_2[x, x]$

```
$TD = 2; (R_{1,2} R_{3,4} // S_{-1} // dm_{1,3→1} // dm_{2,4→2}) // ExpandAB
```

$U_{-2}[] U_{-1}[] U_1[] U_2[] + 2 \alpha \hbar^2 U_{-2}[] U_{-1}[b, y] U_1[] U_2[x] + \frac{1}{2} \hbar^2 U_{-2}[] U_{-1}[y, y] U_1[] U_2[x, x] - \frac{1}{2} e^{-\alpha \beta \hbar} \hbar^2 U_{-2}[] U_{-1}[y, y] U_1[] U_2[x, x]$

```
$TD = 1; S_{-1}[R_{1,2}] ** R_{1,2}
```

$U_{-2}[] U_{-1}[] U_1[] U_2[] + \hbar U_{-2}[] U_{-1}[y] U_1[] U_2[x] - \hbar U_{-2}[] U_{-1}[B[-1], y] U_1[] U_2[x]$

Import from older versions and upgrade/verify!